





**GALAHAD** GLRT

**USER DOCUMENTATION** 

GALAHAD Optimization Library version 3.0

## 1 SUMMARY

Given real n by n symmetric matrices  $\mathbf{H}$  and  $\mathbf{M}$  (with  $\mathbf{M}$  positive definite), real n vectors  $\mathbf{c}$  and  $\mathbf{o}$ , and scalars  $\sigma \ge 0$ ,  $\epsilon \ge 0$  and  $f_0$ , this package finds an **approximate minimizer of the regularised objective function** 

$$\frac{1}{2}\mathbf{x}^T\mathbf{H}\mathbf{x} + \mathbf{c}^T\mathbf{x} + f_0 + \frac{1}{p}\mathbf{\sigma}[\|\mathbf{x} + \mathbf{o}\|_{\mathbf{M}}^2 + \varepsilon]^{p/2},$$

where  $\|\mathbf{v}\|_{\mathbf{M}} = \sqrt{\mathbf{v}^T \mathbf{M} \mathbf{v}}$  is the **M**-norm of **v**. This problem commonly occurs as a subproblem in nonlinear optimization calculations involving cubic regularisation. The method may be suitable for large *n* as no factorization of **H** is required. Reverse communication is used to obtain matrix-vector products of the form  $\mathbf{Hz}$ ,  $\mathbf{M}^{-1}\mathbf{z}$  and, perhaps,  $\mathbf{Mz}$ .

ATTRIBUTES — Versions: GALAHAD\_GLRT\_single, GALAHAD\_GLRT\_double. Uses: GALAHAD\_SYMBOLS, GALAHAD\_SPACE, GALAHAD\_RAND, GALAHAD\_NORMS, GALAHAD\_GLTR, GALAHAD\_ROOTS, GALAHAD\_SPECFILE, \*TTRF. Date: November 2007. Origin: N. I. M. Gould, Oxford University and Rutherford Appleton Laboratory. Language: Fortran 95 + TR 15581 or Fortran 2003.

## 2 HOW TO USE THE PACKAGE

Access to the package requires a USE statement such as

Single precision version

USE GALAHAD\_GLRT\_single

Double precision version

USE GALAHAD\_GLRT\_double

If it is required to use both modules at the same time, the derived types GLRT\_control\_type, GLRT\_inform\_type, GLRT\_data\_type, (Section 2.1) and the subroutines GLRT\_initialize, GLRT\_solve, GLRT\_terminate (Section 2.2) and GLRT\_read\_specfile (Section 2.6) must be renamed on one of the USE statements.

# 2.1 The derived data types

Three derived data types are accessible from the package.

# 2.1.1 The derived data type for holding control parameters

The derived data type GLRT\_control\_type is used to hold controlling data. Default values may be obtained by calling GLRT\_initialize (see Section 2.2.1). The components of GLRT\_control\_type are:

error is a scalar variable of type default INTEGER, that holds the stream number for error messages. Printing of error messages in GLRT\_solve and GLRT\_terminate is suppressed if  $error \le 0$ . The default is error = 6.

out is a scalar variable of type default INTEGER, that holds the stream number for informational messages. Printing of informational messages in GLRT\_solve is suppressed if out < 0. The default is out = 6.

All use is subject to the conditions of the GNU Lesser General Public License version 3. See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

GALAHAD GLRT (March 24, 2018) 1

GLRT GALAHAD

print\_level is a scalar variable of type default INTEGER, that is used to control the amount of informational output which is required. No informational output will occur if print\_level \( \le 0 \). If print\_level \( = 1 \) a single line of output will be produced for each iteration of the process. If print\_level \( \ge 2 \) this output will be increased to provide significant detail of each iteration. The default is print\_level \( = 0 \).

- itmax is a scalar variable of type default INTEGER, that holds the maximum number of iterations which will be allowed in GLRT\_solve. If itmax is set to a negative number, it will be reset by GLRT\_solve to n. The default is itmax = -1.
- extra\_vectors is a scalar variable of type default INTEGER, that specifies the number of additional vectors of length *n* that will be allocated to try to speed up the computation during the second pass. The default is extra\_vectors = 0
- stopping\_rule is a scalar variable of type default INTEGER, that flags the stopping rule to be used (see stop\_relative and stop\_absolute below). Appropriate values are in the range [0,2], and any value outside this range will be interpreted as 0. The default is stopping\_rule = 0.
- freq is a scalar variable of type default INTEGER, that defines the frequency at which the tridiagonal subproblem will be solved. Specifically, the subproblem will be solved on iterations k = 1 + freq \*i, for  $i = 1, 2, \ldots$  If itmax is  $\leq 1$ , it will be reset by GLRT\_solve to 1. The default is freq = 1.
- unitm is a scalar variable of type default LOGICAL, that must be set .TRUE. if the matrix **M** is the identity matrix, and .FALSE. otherwise. The default is unitm = .TRUE.
- space\_critical is a scalar variable of type default LOGICAL, that may be set .TRUE. if the user wishes the package to allocate as little internal storage as possible, and .FALSE. otherwise. The package may be more efficient if space\_critical is set .FALSE.. The default is space\_critical = .FALSE..
- deallocate\_error\_fatal is a scalar variable of type default LOGICAL, that may be set .TRUE. if the user wishes the package to return to the user in the unlikely event that an internal array deallocation fails, and .FALSE. if the package should be allowed to try to continue. The default is deallocate\_error\_fatal = .FALSE..
- stop\_relative and stop\_absolute are scalar variables of type REAL (double precision in GALAHAD\_GLRT\_double), that holds the relative and absolute convergence tolerances (see Section 4). The computed solution  $\mathbf{x}$  is accepted by GLRT\_solve if the computed value of  $\|\mathbf{H}\mathbf{x} + \lambda \mathbf{M}\mathbf{x} + \mathbf{c}\|_{\mathbf{M}^{-1}}$  is less than or equal to  $\max(\mathbf{v}\|\mathbf{c}\|_{\mathbf{M}^{-1}}*$  stop\_relative, stop\_absolute), where  $\lambda = \sigma[\|\mathbf{x} + \mathbf{o}\|_{\mathbf{M}}^2 + \mathbf{c}]^{p/2-1}$  and  $\mathbf{v}$  depends on the stopping rule selected by stopping\_rule: for stopping\_rule = 0,  $\mathbf{v} = 1$ , for stopping\_rule = 1,  $\mathbf{v} = \min(1, \|\mathbf{x}\|)$ , and for stopping\_rule = 2,  $\mathbf{v} = \min(1, \|\mathbf{x}\|/\max(1, \sigma))$ . The defaults are stop\_relative =  $\sqrt{u}$  and stop\_absolute = 0.0, where u is EPSILON(1.0) (EPSILON(1.0D0) in GALAHAD\_GLRT\_double).
- fraction\_opt is a scalar variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that specifies the fraction of the optimal value which is to be considered acceptable by the algorithm. A negative value is considered to be zero, and a value of larger than one is considered to be one. Reducing fraction\_opt below one will result in a reduction of the computation performed at the expense of an inferior optimal value. The default is fraction\_opt = 1.0.
- f\_0 is a scalar variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that gives the value of the constant term  $f_0$  in the objective function. This value has no effect on the computed minimizer  $\mathbf{x}$ . The default is  $f_0 = 0.0$ .
- rminvr\_zero is a scalar variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that gives the smallest value that the square of the M-norm of the gradient of the objective function may be before it is considered to be zero. The default is rminvr\_zero = 10 u, where u is EPSILON(1.0) (EPSILON(1.0D0) in GALAHAD\_GLRT\_double).

prefix is a scalar variable of type default CHARACTER and length 30, that may be used to provide a user-selected character string to preface every line of printed output. Specifically, each line of output will be prefaced by the string prefix (2:LEN(TRIM(prefix))-1), thus ignoring the first and last non-null components of the supplied string. If the user does not want to preface lines by such a string, they may use the default prefix = "".

## 2.1.2 The derived data type for holding informational parameters

The derived data type GLRT\_inform\_type is used to hold parameters that give information about the progress and needs of the algorithm. The components of GLRT\_inform\_type are:

- status is a scalar variable of type default INTEGER, that gives the current status of the algorithm. See Sections 2.3 and 2.4 for details.
- alloc\_status is a scalar variable of type default INTEGER, that gives the status of the last internal array allocation or deallocation. This will be 0 if status = 0.
- bad\_alloc is a scalar variable of type default CHARACTER and length 80, that gives the name of the last internal array for which there were allocation or deallocation errors. This will be the null string if status = 0.
- obj is a scalar variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that holds the value of the quadratic function  $\frac{1}{2}\mathbf{x}^T\mathbf{H}\mathbf{x} + \mathbf{c}^T\mathbf{x} + f_0$ .
- obj\_regularized is a scalar variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that holds the value of the regularized objective function  $\frac{1}{2}\mathbf{x}^T\mathbf{H}\mathbf{x} + \mathbf{c}^T\mathbf{x} + f_0 + \frac{1}{p}\sigma[\|\mathbf{x} + \mathbf{o}\|_{\mathbf{M}}^2 + \epsilon]^{p/2}$ .
- multiplier is a scalar variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that holds the value of the multiplier  $\lambda = \sigma[\|\mathbf{x} + \mathbf{o}\|_{\mathbf{M}}^2 + \epsilon]^{p/2-1}$ .
- leftmost is a scalar variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that holds an estimate of the leftmost eigenvalue of the matrix pencil  $(\mathbf{H}, \mathbf{M})$ .
- iter is a scalar variable of type default INTEGER, that holds the current number of Lanczos vectors used.
- iter\_pass2 is a scalar variable of type default INTEGER, that holds the current number of Lanczos vectors used in the second pass.
- negative\_curvature is a scalar variable of type default LOGICAL, that is set .TRUE. if **H** has been found to be indefinite during the calculation and .FALSE. otherwise.

# 2.1.3 The derived data type for holding problem data

The derived data type GLRT\_data\_type is used to hold all the data for a particular problem between calls of GLRT procedures. This data should be preserved, untouched, from the initial call to GLRT\_initialize to the final call to GLRT\_terminate.

## 2.2 Argument lists and calling sequences

There are three procedures for user calls (see Section 2.6 for further features):

- 1. The subroutine GLRT\_initialize is used to set default values, and initialize private data.
- 2. The subroutine GLRT\_solve is called repeatedly to solve the problem. On each exit, the user may be expected to provide additional information and, if necessary, re-enter the subroutine.

GLRT GALAHAD

3. The subroutine GLRT\_terminate is provided to allow the user to automatically deallocate array components of the private data, allocated by GLRT\_solve, at the end of the solution process. It is important to do this if the data object is re-used for another problem since GLRT\_initialize cannot test for this situation, and any existing associated targets will subsequently become unreachable.

We use square brackets [ ] to indicate OPTIONAL arguments.

## 2.2.1 The initialization subroutine

Default values are provided as follows:

```
CALL GLRT_initialize( data, control, inform )
```

- data is a scalar INTENT (INOUT) argument of type GLRT\_data\_type (see Section 2.1.3). It is used to hold data about the problem being solved.
- control is a scalar INTENT (OUT) argument of type GLRT\_control\_type (see Section 2.1.1). On exit, control contains default values for the components as described in Section 2.1.1. These values should only be changed after calling GLRT\_initialize.
- inform is a scalar INTENT (OUT) argument of type GLRT\_inform\_type (see Section 2.1.2). A successful call to GLRT\_initialize is indicated when the component status has the value 0. For other return values of status, see Section 2.4.

# 2.2.2 The optimization problem solution subroutine

The optimization problem solution algorithm is called as follows:

```
CALL GLRT_solve( n, p, sigma, X, R, VECTOR, data, control, inform[, eps, O] )
```

- is a scalar INTENT (IN) argument of type default INTEGER, that must be set to the number of unknowns, n. **Restriction:** n > 0.
- is a scalar INTENT (IN) variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that must be set on initial entry to the desired order p of regularisation. **Restriction:**  $p \ge 2$ .
- sigma is a scalar INTENT (IN) variable of type default REAL (double precision in GALAHAD\_GLRT\_double), that must be set on initial entry to the value of the weight  $\sigma$  associated with the regularisation term. **Restriction:**  $\sigma > 0$ .
- is an array INTENT (INOUT) argument of dimension n and type default REAL (double precision in GALAHAD\_GL-RT\_double), that holds an estimate of the solution x of the linear system. On initial entry, X need not be set. It must not be changed between entries. On exit, X contains the current best estimate of the solution.
- is an array INTENT (INOUT) argument of dimension n and type default REAL (double precision in GALAHAD\_GL-RT\_double), that is used to hold the gradient  $\mathbf{H}\mathbf{x} + \mathbf{c}$  of the objective function at the current estimate of the solution. On initial entry, R must contain the vector  $\mathbf{c}$ . If inform%status = 5 on exit, it must be reset to  $\mathbf{c}$ ; otherwise it must be left unchanged. On exit, R contains the gradient of the objective function at the current best estimate of the solution.
- VECTOR is an array INTENT(INOUT) argument of dimension n and type default REAL (double precision in GALAH-AD\_GLRT\_double), that is used to pass information from and to GLRT\_solve, as explained in Section 2.3. On initial entry, VECTOR need not be set. On exit, the actual content of the array depends on the value of the parameter inform%status (see Section 2.3).
- data is a scalar INTENT (INOUT) argument of type GLRT\_data\_type (see Section 2.1.3). It is used to hold data about the problem being solved. It must not have been altered by the user since the last call to GLRT\_initialize.



control is a scalar INTENT (IN) argument of type GLRT\_control\_type. (see Section 2.1.1). Default values may be assigned by calling GLRT\_initialize prior to the first call to GLRT\_solve.

- inform is a scalar INTENT (INOUT) argument of type GLRT\_inform\_type (see Section 2.1.2). On initial entry, the component status must be set to 1. The remaining components need not be set. A successful call to GL-RT\_solve is indicated when the component status has the value 0. For other return values of status, see Sections 2.3 and 2.4.
- is an optional scalar INTENT (IN) variable of type default REAL (double precision in GALAHAD\_GLRT\_double). If eps is PRESENT, it must be set on initial entry to the value of the shift ε associated with the regularisation term. **Restriction:**  $\varepsilon > 0$ .
- is an optional array INTENT (IN) variable of type default REAL (double precision in GALAHAD\_GLRT\_double). If 0 is PRESENT, it must be set on initial entry to the value of the offset o associated with the regularisation term.

#### 2.2.3 The termination subroutine

All previously allocated arrays are deallocated as follows:

data is a scalar INTENT(INOUT) argument of type GLRT\_data\_type exactly as for GLRT\_solve that must not have been altered by the user since the last call to GLRT\_initialize. On exit, array components will have been deallocated.

control is a scalar INTENT (IN) argument of type GLRT\_control\_type exactly as for GLRT\_solve.

inform is a scalar INTENT (OUT) argument of type GLRT\_type exactly as for GLRT\_solve. Only the component status will be set on exit, and a successful call to GLRT\_terminate is indicated when this component status has the value 0. For other return values of status, see Section 2.4.

# 2.3 Reverse communication

A positive value of inform%status on exit from GLRT\_solve indicates that the user needs to take appropriate action before re-entering the subroutine. Possible values are:

2. The user must perform the preconditioning operation

$$\mathbf{y} := \mathbf{M}^{-1}\mathbf{z},$$

and recall GLRT\_solve. The vector **z** is available in the array VECTOR, and the result **y** must be placed in VECTOR. No argument except VECTOR should be altered before recalling GLRT\_solve. This return can only occur when control%unitm is .FALSE..

3. The user must perform the matrix-vector product

$$\mathbf{y} := \mathbf{H}\mathbf{z}$$

and recall GLRT\_solve. The vector **z** is available in the array VECTOR, and the result **y** must be placed in VECTOR. No argument except VECTOR should be altered before recalling GLRT\_solve.

- 4. The user should reset R to c and recall GLRT\_solve. No argument except R should be altered before recalling GLRT\_solve.
- 5. The user must perform the operation

$$y := Mz$$

and recall GLRT\_solve. The vector **z** is available in the array VECTOR, and the result **y** must be placed in VECTOR. No argument except VECTOR should be altered before recalling GLRT\_solve. This return can only occur when both control%unitm is .FALSE, and O is PRESENT.

GLRT GALAHAD

### 2.4 Warning and error messages

A negative value of inform%status on exit from GLRT\_solve or GLRT\_terminate indicates that an error has occurred. No further calls should be made until the error has been corrected. Possible values are:

- -1. An allocation error occurred. A message indicating the offending array is written on unit control%error, and the returned allocation status and a string containing the name of the offending array are held in inform%alloc\_status and inform%bad\_alloc respectively.
- -2. A deallocation error occurred. A message indicating the offending array is written on unit control%error and the returned allocation status and a string containing the name of the offending array are held in inform%alloc\_status and inform%bad\_alloc respectively.
- -3. (GLRT\_solve only) One or more of the restrictions n > 0, sigma  $\geq 0$ , eps  $\geq 0$  or  $p \geq 2$  has been violated.
- -7. (GLRT\_solve only) The problem is unbounded from below. This can only happen if  $p \le 2$ . In this case, the problem is unbounded along the arc X  $+\alpha$  VECTOR as  $\alpha$  increases.
- -15. (GLRT\_solve only) The matrix **M** appears not to be positive definite.
- -18. (GLRT\_solve only) More than control%itmax iterations have been performed without obtaining convergence.
- -25. (GLRT\_solve only) inform% status is not > 0 on entry.

# 2.5 Re-entry with a new value of $\sigma$

It commonly happens that, having solved the problem for a particular value of the weight  $\sigma$ , a user now wishes to solve the problem for a different value of  $\sigma$ . Rather than restarting the calculation with inform%status = 1, a useful approximation may be found resetting sigma to the new required value and R to c, and recalling GLRT\_solve with inform%status = 6 and the remaining arguments unchanged. This will determine the best solution within the Krylov space investigated in the previous minimization (see Section 4).

# 2.6 Further features

In this section, we describe an alternative means of setting control parameters, that is components of the variable control of type GLRT\_control\_type (see Section 2.1.1), by reading an appropriate data specification file using the subroutine GLRT\_read\_specfile. This facility is useful as it allows a user to change GLRT control parameters without editing and recompiling programs that call GLRT.

A specification file, or specfile, is a data file containing a number of "specification commands". Each command occurs on a separate line, and comprises a "keyword", which is a string (in a close-to-natural language) used to identify a control parameter, and an (optional) "value", which defines the value to be assigned to the given control parameter. All keywords and values are case insensitive, keywords may be preceded by one or more blanks but values must not contain blanks, and each value must be separated from its keyword by at least one blank. Values must not contain more than 30 characters, and each line of the specfile is limited to 80 characters, including the blanks separating keyword and value.

The portion of the specification file used by GLRT\_read\_specfile must start with a "BEGIN GLRT" command and end with an "END" command. The syntax of the specifile is thus defined as follows:

```
( .. lines ignored by GLRT_read_specfile .. )
BEGIN GLRT
   keyword value
   .....
keyword value
```

```
END
( .. lines ignored by GLRT_read_specfile .. )
```

where keyword and value are two strings separated by (at least) one blank. The "BEGIN GLRT" and "END" delimiter command lines may contain additional (trailing) strings so long as such strings are separated by one or more blanks, so that lines such as

```
BEGIN GLRT SPECIFICATION

and

END GLRT SPECIFICATION
```

are acceptable. Furthermore, between the "BEGIN GLRT" and "END" delimiters, specification commands may occur in any order. Blank lines and lines whose first non-blank character is ! or \* are ignored. The content of a line after a ! or \* character is also ignored (as is the ! or \* character itself). This provides an easy manner to "comment out" some specification commands, or to comment specific values of certain control parameters.

The value of a control parameters may be of three different types, namely integer, logical or real. Integer and real values may be expressed in any relevant Fortran integer and floating-point formats (respectively). Permitted values for logical parameters are "ON", "TRUE", ".TRUE.", "T", "YES", "Y", or "OFF", "NO", "N", "FALSE", ".FALSE." and "F". Empty values are also allowed for logical control parameters, and are interpreted as "TRUE".

The specification file must be open for input when <code>GLRT\_read\_specfile</code> is called, and the associated device number passed to the routine in device (see below). Note that the corresponding file is <code>REWINDed</code>, which makes it possible to combine the specifications for more than one program/routine. For the same reason, the file is not closed by <code>GLRT\_read\_specfile</code>.

# 2.6.1 To read control parameters from a specification file

Control parameters may be read from a file as follows:

```
CALL GLRT read specfile (control, device)
```

control is a scalar INTENT (INOUT) argument of type GLRT\_control\_type (see Section 2.1.1). Default values should have already been set, perhaps by calling GLRT\_initialize. On exit, individual components of control may have been changed according to the commands found in the specifile. Specifile commands and the component (see Section 2.1.1) of control that each affects are given in Table 2.1.

device is a scalar INTENT (IN) argument of type default INTEGER, that must be set to the unit number on which the specifle has been opened. If device is not open, control will not be altered and execution will continue, but an error message will be printed on unit control%error.

# 2.7 Information printed

If control\*print\_level is positive, information about the progress of the algorithm will be printed on unit control\*out. If control\*print\_level = 1, a single line of output will be produced for each iteration of the process. This will include the iteration number, the value of the objective function, the  $\mathbf{M}^{-1}$ -norm of its gradient  $\|(\mathbf{H} + \lambda \mathbf{M})\mathbf{x} + \mathbf{c}\|_{\mathbf{M}^{-1}}$ , the value of the multiplier  $\lambda = \sigma[\|\mathbf{x} + \mathbf{o}\|_{\mathbf{M}}^2 + \epsilon]^{p/2-1}$ , the number of Newton steps required to find  $\lambda$ , and the exit code from this calculation (0 = successful, 1 = 1, stalled, 2 = more than 100 steps). If control\*print\_level  $\geq$  2, this output will be increased to provide significant detail of each iteration. This extra output includes a complete history of the inner iteration required to solve the "tridiagonal" subproblem, and for each Newton iteration records the estimate of  $\lambda$ , the error  $\theta(\lambda) = [\|\mathbf{x} + \mathbf{o}\|_{\mathbf{M}}^2 + \epsilon]^{p/2-1} - \lambda/\sigma$  and the  $\mathbf{M}$ -norm of  $\mathbf{x}$ .

command	component of control	value type
error-printout-device	%error	integer
printout-device	%out	integer
print-level	%print_level	integer
maximum-number-of-iterations	%itmax	integer
number-extra-n-vectors-used	%extra_vectors	integer
stopping-rule	%stopping_rule	integer
tri-diagonal-solve-frequency	%freq	integer
relative-accuracy-required	%stop_relative	real
absolute-accuracy-required	%stop_absolute	real
fraction-optimality-required	%fraction_opt	real
constant-term-in-objective	%f_0	real
zero-gradient-tolerance	%rminvr_zero	real
two-norm-regularisation	%unitm	logical
space-critical	%space_critical	logical
deallocate-error-fatal	%deallocate_error_fatal	logical

Table 2.1: Specifle commands and associated components of control.

# **3 GENERAL INFORMATION**

Use of common: None.

**Workspace:** Provided automatically by the module.

Other routines called directly: GLRT\_solve calls the LAPACK subroutine \*PTTRF, where \* is S for the default real version and D for the double precision version.

Other modules used directly: GLRT\_solve calls the GALAHAD packages GALAHAD\_SYMBOLS, GALAHAD\_SPACE, GALAHAD\_RAND, GALAHAD\_NORMS, GALAHAD\_GLTR, GALAHAD\_ROOTS and GALAHAD\_SPECFILE.

Input/output: Output is under control of the arguments control%error, control%out and control%print\_level.

**Restrictions:** n > 0,  $\sigma > 0$ .

**Portability:** ISO Fortran 95 + TR 15581 or Fortran 2003. The package is thread-safe.

#### 4 METHOD

The required solution  $\mathbf{x}$  necessarily satisfies the optimality condition  $\mathbf{H}\mathbf{x} + \lambda\mathbf{M}\mathbf{x} + \mathbf{c} + \lambda\mathbf{o} = 0$ , where  $\lambda = \sigma[\|\mathbf{x} + \mathbf{o}\|_{\mathbf{M}}^2 + \epsilon]^{p/2-1}$ . In addition, the matrix  $\mathbf{H} + \lambda\mathbf{M}$  will be positive semi-definite.

The method is iterative. Starting with the vector  $\mathbf{M}^{-1}\mathbf{c}$ , a matrix of Lanczos vectors is built one column at a time so that the k-th column is generated during iteration k. These columns span a so-called Krylov space. The resulting n by k matrix  $\mathbf{Q}_k$  has the property that  $\mathbf{Q}_k^T \mathbf{H} \mathbf{Q}_k = \mathbf{T}_k$ , where  $\mathbf{T}_k$  is tridiagonal. An approximation to the required solution may then be expressed formally as

$$\mathbf{x}_{k+1} = \mathbf{Q}_k \mathbf{y}_k$$

where  $\mathbf{y}_k$  solves the "tridiagonal" subproblem of minimizing

$$\frac{1}{2}\mathbf{y}^{T}\mathbf{T}_{k}\mathbf{y} + \|\mathbf{c}\|_{\mathbf{M}^{-1}}\mathbf{e}_{1}^{T}\mathbf{y} + \frac{1}{p}\boldsymbol{\sigma}[\|\mathbf{y} + \mathbf{d}\|_{2}^{2} + \boldsymbol{\varepsilon}]^{p/2},\tag{4.1}$$

where  $\mathbf{d} = \mathbf{Q}_k^T \mathbf{Mo}$  and  $\mathbf{e}_1$  is the first unit vector.

To minimize (4.1), the optimality conditions

$$(\mathbf{T}_k + \lambda \mathbf{I})\mathbf{y}(\lambda) = -\mathbf{c} - \lambda \mathbf{d}, \tag{4.2}$$

where  $\lambda = \sigma[\|\mathbf{y}(\lambda) + \mathbf{d}\|_{\mathbf{M}}^2 + \varepsilon]^{p/2-1}$  are used as the basis of an iteration. Specifically, given an estimate  $\lambda$  for which  $\mathbf{T}_k + \lambda \mathbf{I}$  is positive definite, the tridiagonal system (4.2) may be efficiently solved to give  $\mathbf{y}(\lambda)$ . It is then simply a matter of adjusting  $\lambda$  (for example by a Newton-like process) to solve the scalar nonlinear equation

$$\theta(\lambda) \equiv [\|\mathbf{y}(\lambda) + \mathbf{d}\|_{\mathbf{M}}^2 + \varepsilon]^{p/2 - 1} - \frac{\lambda}{\sigma} = 0.$$
(4.3)

In practice (4.3) is reformulated, and a more rapidly converging iteration is used.

It is possible to measure the optimality measure  $\|\mathbf{H}\mathbf{x} + \lambda \mathbf{M}\mathbf{x} + \mathbf{c} + \lambda \mathbf{o}\|_{\mathbf{M}^{-1}}$  without computing  $\mathbf{x}_{k+1}$ , and thus without needing  $\mathbf{Q}_k$ . Once this measure is sufficiently small, a second pass is required to obtain the estimate  $\mathbf{x}_{k+1}$  from  $\mathbf{y}_k$ . As this second pass is an additional expense, a record is kept of the optimal objective function values for each value of k, and the second pass is only performed so far as to ensure a given fraction of the final optimal objective value. Large savings may be made in the second pass by choosing the required fraction to be significantly smaller than one.

Special code is used in the special case p = 2, as in this case a single pass suffices.

#### **Reference:** The method is described in detail in

C. Cartis, N. I. M. Gould and Ph. L. Toint, Adaptive cubic regularisation methods for unconstrained optimization. Part I: motivation, convergence and numerical results. *Mathematical Programming* **127(2)**, pp.245-295, 2011.

# 5 EXAMPLE OF USE

Suppose we wish to solve a problem in 10,000 unknowns, whose data is

$$\mathbf{H} = \begin{pmatrix} -2 & 1 & & & \\ 1 & -2 & . & & \\ & . & . & . & \\ & & . & -2 & 1 \\ & & & 1 & -2 \end{pmatrix}, \ \mathbf{M} = \begin{pmatrix} 2 & & & & \\ & 2 & & & \\ & & . & & \\ & & & 2 & \\ & & & & 2 \end{pmatrix}, \ \mathbf{c} = \begin{pmatrix} 1 \\ 1 \\ . \\ 1 \\ 1 \end{pmatrix} \text{ and } \mathbf{o} = -\begin{pmatrix} 1 \\ 1 \\ . \\ 1 \\ 1 \end{pmatrix},$$

with a weight  $\sigma = 10$  and shift  $\varepsilon = 1$ . Suppose further that we are content with an approximation which is within 99% of the best. Then we may use the following code

```
PROGRAM GALAHAD_GLRT_EXAMPLE ! GALAHAD 2.7 - 08/02/2016 AT 09:50 GMT.
USE GALAHAD_GLRT_DOUBLE
                                               ! double precision version
IMPLICIT NONE
INTEGER, PARAMETER :: working = KIND( 1.0D+0 ) ! set precision
REAL (KIND = working), PARAMETER :: one = 1.0_working, two = 2.0_working
INTEGER, PARAMETER :: n = 10000
                                               ! problem dimension
INTEGER :: i
REAL (KIND = working) :: p = 3.0_working
                                               ! order of regulatisation
REAL (KIND = working) :: eps = 1.0 working ! shift
REAL (KIND = working) :: sigma = 10.0_working! regulatisation weight
REAL ( KIND = working ), DIMENSION( n ) :: X, R, VECTOR, H_vector, O
TYPE ( GLRT_data_type ) :: data
TYPE ( GLRT_control_type ) :: control
TYPE ( GLRT_inform_type ) :: inform
CALL GLRT_initialize( data, control, inform ) ! Initialize control parameters
```

All use is subject to the conditions of the GNU Lesser General Public License version 3. See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

GALAHAD GLRT (March 24, 2018) 9

**GLRT** 

```
control%unitm = .FALSE.
                                     ! M is not the identity matrix
   control%fraction_opt = 0.99
                                     ! Only require 99% of the best
  R = one
                                      ! The linear term c is a vector of ones
  0 = - one
                                      ! The offset o is a vector of minus ones
   inform%status = 1
  DO
                                       ! Iteration to find the minimizer
    CALL GLRT_solve( n, p, sigma, X, R, VECTOR, data, control, inform,
                    eps = eps, 0 = 0)
    SELECT CASE( inform%status )
                                      ! Branch as a result of inform%status
    CASE (2)
                                      ! Form the preconditioned vector
      VECTOR = VECTOR / two
                                      ! Preconditioner is two times identity
                                      ! Form the matrix-vector product
    CASE ( 3 )
      H_{vector}(1) = - two * VECTOR(1) + VECTOR(2)
      DO i = 2, n - 1
       H_{vector}(i) = VECTOR(i-1) - two * VECTOR(i) + VECTOR(i+1)
      END DO
      H_{vector}(n) = VECTOR(n-1) - two * VECTOR(n)
      VECTOR = H_vector
    CASE (4) ! Restart
      R = one
                     ! set r to c
    CASE (5)
                                      ! Form the product of the preconditioner
      VECTOR = two * VECTOR
                                      ! with a vector
    CASE ( 0 ) ! Successful return
      WRITE( 6, "( 1X, I0, ' 1st pass and ', I0, ' 2nd pass iterations' )" ) &
       inform%iter, inform%iter_pass2
      H_{vector}(1) = - two * X(1) + X(2)
      DO i = 2, n - 1
        H_{vector}(i) = X(i-1) - two * X(i) + X(i+1)
      END DO
      H_{vector}(n) = X(n-1) - two * X(n)
      WRITE( 6, "( ' objective recurred and calculated = ', 2ES16.8 )" )
                                                                            &
       inform%obj_regularized, 0.5_working * DOT_PRODUCT( X, H_vector ) +
                                                                            δ
         SUM( X ) + ( sigma / p ) * ( two * DOT_PRODUCT( X + 0, X + 0 ) +
           eps ) ** ( p/two )
      CALL GLRT_terminate( data, control, inform ) ! delete internal workspace
      EXIT
                    ! Error returns
    CASE DEFAULT
      WRITE( 6, "( ' GLRT_solve exit status = ', I6 ) " ) inform%status
      CALL GLRT_terminate( data, control, inform ) ! delete internal workspace
      EXIT
    END SELECT
   END PROGRAM GALAHAD_GLRT_EXAMPLE
This produces the following output:
```

```
3 1st pass and 1 2nd pass iterations
objective recurred and calculated = 9.88721600E+03 9.88721600E+03
```